

FP&A プロフェッショナルに要請される スプレッドシートのスキル大系

— 基本操作と主要な関数に着目して —

横井 隆志

1. はじめに

1. スプレッドシートの役割

スプレッドシートは、定量分析のための基本ツールとして活用されるものであり、FP&A の業務においても重要なツールの 1 つであるといえます。スプレッドシートの代表的アプリケーションである Excel は 1985 年にリリースされており、この Excel というスプレッドシートアプリケーションの登場が FP&A の進化に大きな影響を与えたとされています¹。

FP&A の業務では、データの整理、計算、視覚化が必要です。これらの作業に対応するために、スプレッドシートは必須のツールといえるでしょう。もちろん、現在では BI ツールや会計ソフトなど、他にも便利なツールがたくさんあります。しかし、それらのツールから得られる情報を横断的に加工したり、活用したりするにはスプレッドシートの存在が欠かせません。そのため、FP&A プロフェッショナルの必須スキルの一つといえます。

本稿では、Association for Financial Professionals, 2024, *FPAC Exam Prep Platform: Chapter 15 Spreadsheet Functions*. を紐解きながら、FP&A プロフェッショナルにどのようなスキルが要請されるのかを、スプレッドシートの基本操作と主要な関数に着目して明らかにしていきます。

2. データ活用の現状と課題

2024 年 7 月に FP&A のプロフェッショナルの方々と意見交換する機会がありました。その際に、データの活用に関する課題が多くの方から指摘されました。具体的には、「目の前に大量のデータはあるが、それを意思決定や情報共有に活用しきれていない」という声が多かったのです。

この課題を解決するための一つの手段が、BI ツールの活用です。BI ツールを使えば、より高度な可視化が可能になりますが、BI ツールを活用する前提としても、基本的な汎用ツールとしてのスプレッドシートは重要です。何より、Excel を用いて簡単な加工や修正を行うだけで、他者に情報をわかりやすく共有することができるという利点があります。

¹ 2024 年 9 月 10 日に実施した FPAC オンライン集中講座で、本学特任教授の石橋 善一郎先生がこのことをご教示くださったときに目から鱗が落ちました。石橋先生に感謝申し上げます。

3. スプレッドシートを活用するメリット

FP&A プロフェッショナルがスプレッドシートを使用する具体的な利点について見ていきましょう。

1つ目の利点は、計算の正確性が向上することです。スプレッドシートでは、数式を入力するだけで自動的に計算が行われるため、手作業による計算ミスを防ぐことができます。

2つ目は、検証が容易になる、ということです。スプレッドシートのセルの参照を追跡すれば、計算ロジックを簡単に確認できるため、データの検証が容易です。

3つ目は自動再計算機能です。データを修正すると、自動的に計算結果が更新されるため、作業の生産性が向上します。

4つ目は財務モデルの構築が容易になる、ということです。スプレッドシートは、財務モデルの構築を簡便化し、データ活用のハードルを下げてください。

II. 数式の入力と参照

それではここから、Excel の基本操作を確認していきましょう。

1. 数値と文字列

Excel のワークシートはデータを格納できる複数のセルによって構成され、セルの位置は列番号 (A, B, C…) と行番号 (1, 2, 3…) で表されるセル番地で認識されます。セルには、直接データを入力することも、数式を入力することもできます。これらのデータは数値と文字列に大別され、両者の違いは、数値は計算が可能であり、文字列は計算ができない、という点にあります。

2. オートフィルの活用

オートフィルは、アクティブセルの右下にある「フィルハンドル」をドラッグすることで、データや数式を一括して入力する機能です。数値や文字列、数式などを入力したセルの右下隅にカーソルを合わせると、黒い十字マークに変わります。この状態でドラッグすることで、水平方向、垂直方向へデータをコピーできます。オートフィルには、以下の機能が備わっています。

① 数値の連続入力

例えば、セル A1 に 1 を入力し、A2 に 2 を入力した状態で 2 つのセルを選択してドラッグすると、3, 4, 5 のように連続した数値が自動入力されます。

② パターンの自動生成

曜日 (例: 月, 火, 水) や月名 (例: 1 月, 2 月, 3 月) など、Excel が認識可能なデータパターンを使用すると、オートフィルで簡単に補完できます。

③ オートフィルオプション

ドラッグ後に表示される「オートフィルオプション」を使用すると、コピー、連続データ生成、書式のみコピーなど、目的に応じた適用が可能です。

④ 数式の適用

数式を含むセルをオートフィルすると、同じ構造の数式を他のセルに複写することができます。

3. 相対参照と絶対参照

(1) 相対参照

数式をオートフィルする際、理解しておく必要があるのが、相対参照と絶対参照、すなわち、Excel がどのようにセルを参照しているのか、ということです。

通常、セルの参照は数式が入力されているセルからの相対的な位置、つまり、どちら方向へ、何行、何列離れているか、で認識されています。

(2) 絶対参照・複合参照

一方、参照先を固定したいときには、「\$D\$17」のように、列番号や行番号の前に「\$」記号をつけます。参照方法を切り替えるには F4 キーを使います。数式の中で参照方法を切り替えたいセル番地を選択して F4 キーを押すと、相対参照 (D17) から列と行が両方固定された絶対参照 (\$D\$17)、行のみ固定された複合参照 (D\$17)、列のみ固定された複合参照 (\$D17)、の 4 つが順番に切り替わります。

図 1. オートフィルの活用例

	A	B	C	D	E
1	初期投資	¥1,500,000			
2					
3		成長率			
4	年	3%	5%	7%	10%
5	1	=B\$1*(1+B\$4)^\$A5			
6	2				
7	3				
8	4				
9	5				
10	6				
11	7				
12	8				
13	9				
14	10				

出所：AFP(2024) Exhibit15-2 を参考に筆者作成

図 1 では、B1 に入力された初期投資の金額をもとに、4 段階の成長率で 1~10 年が経過した後の金額を試算します。このとき、初期投資額が参照されている B1 は絶対参照に、成長率を参照している B4 は行のみ、年を参照している A5 は列のみを固定することで、水平方向、垂直方向への二段階のオートフィルで表を埋めることが可能になります。

図 2. オートフィル後の完成イメージ

E14 : × ✓ fx =-\$B\$1*(1+E\$4)^\$A14					
	A	B	C	D	E
1	初期投資	¥1,500,000			
2					
3		成長率			
4	年	3%	5%	7%	10%
5	1	1,545,000	1,575,000	1,605,000	1,650,000
6	2	1,591,350	1,653,750	1,717,350	1,815,000
7	3	1,639,091	1,736,438	1,837,565	1,996,500
8	4	1,688,263	1,823,259	1,966,194	2,196,150
9	5	1,738,911	1,914,422	2,103,828	2,415,765
10	6	1,791,078	2,010,143	2,251,096	2,657,342
11	7	1,844,811	2,110,651	2,408,672	2,923,076
12	8	1,900,155	2,216,183	2,577,279	3,215,383
13	9	1,957,160	2,326,992	2,757,689	3,536,922
14	10	2,015,875	2,443,342	2,950,727	3,890,614

出所：AFP(2024) Exhibit15-2 を参考に筆者作成

III. 主な関数

AFP(2024)では、財務分析等に用いられる基本的な関数として、

- 論理関数（条件分岐による処理を行う IF 関数、AND 関数、OR 関数など、）
- 集計関数（SUMIF、COUNTIF など、条件に基づく合計やカウントを行う関数）
- 検索関数（VLOOKUP、HLOOKUP など、データの検索や抽出を行う関数）
- 財務関数（PV、FV など、将来価値や現在価値、一般的な金融計算を行う関数）

など、多くの関数に着目しています。さらには、会計実務での活用を想定して、複数の関数をネストするなどして複合的に活用する、極めて実用的な方法が具体的に示されているのも AFP(2024)の大きな特徴といえます。

本稿では、財務分析や財務モデルの構築に特に有用な関数として、条件に基づく集計を行う集計関数と、より複雑なシナリオを扱うような場合に、論理関数よりも効率的な作業を可能にする検索関数を取り上げます。また、AFP(2024)で積極的に活用されている配列数式に注目し、具体的な事例とともに解説を試みます。

1. 集計関数

(1) COUNTIF 関数と COUNTIFS 関数

COUNTIF は、条件に一致するセルの数を数える関数です。

引数は以下の様に構成されます。

=COUNTIF(範囲, 条件)

複数の条件に一致するセルの数を数えるには、COUNTIFS 関数を使用します。

=COUNTIFS(条件範囲 1, 条件 1, 条件範囲 2, 条件 2, ...)

(2) SUMIF 関数と SUMIFS 関数

SUMIF 関数は、範囲の中で条件に該当する行もしくは列を抽出し、その行もしくは列の合計範囲に含まれる値を合計します。

=SUMIF(範囲, 条件, 合計範囲)

SUMIFS 関数は、1 つ目の引数として合計範囲を指定し、2 つ目以降の引数に複数の条件範囲と条件をペアで指定することができます。

=SUMIFS(合計範囲, 条件範囲 1, 条件 1, 条件範囲 2, 条件 2, ...)

(3) 集計関数の条件の指定方法

それでは、集計関数の具体的な使用例を見て行きましょう。

図 3. 集計関数の活用例

	A	B	C	D	E	F	G	H
1	月	部署	用途	金額		COUNTIF		
2	11月	営業部	旅費	¥26,800		旅費		5
3	11月	総務部	消耗品費	¥38,800		金額 > 3万		7
4	11月	営業部	旅費	¥54,900		金額 >	30,000	7
5	11月	総務部	会議費	¥34,400				
6	11月	営業部	旅費	¥10,700		SUMIF		
7	11月	営業部	旅費	¥36,500		旅費		163,800
8	11月	営業部	会議費	¥14,000				
9	11月	総務部	消耗品費	¥32,900			COUNTIFS	SUMIFS
10	11月	総務部	旅費	¥34,900		旅費	件数	金額
11	11月	営業部	会議費	¥25,300		営業部	4	128900
12	11月	総務部	消耗品費	¥56,600		総務部	1	34900

出所：AFP(2024) Exhibit15-4 を参考に筆者作成

図 3 のセル H2 は、C2:C12 で用途が旅費に該当する件数を返しています。式は、条件をセルの参照で指定する場合は

=COUNTIF(C2:C12,F2)


条件を直接入力する場合は

=COUNTIF(C2:C12,"旅費")

の様に指定します。


引数に文字列を指定する場合は、半角のダブルクォーテーション「”」で文字列の前後を囲む必要があります。関数を入力する際に数式バーの左の「fx」ボタン（関数の入力ボタン）から関数ウィザードを使用することにより、引数が正常に認識されているかどうかを確認しながら引数の入力を行うことができます。

図 4. 検索条件が認識されていない例



出所：筆者作成

図 5. 検索条件が正常に認識された例



出所：筆者作成

図 3 のセル H3 は、D2:D12 で金額が 3 万円を超える件数を返しています。集計関数で入力された値より大きい、未満、以上、以下といった条件を指定する場合も、比較演算子を使った条件式をダブルクォーテーションで囲む必要があります。セル H3 の式は以下の通りです。

=COUNTIF(D2:D12,">30000")

図 3 のセル H4 では、セル G4 の値より大きい、という、セルを参照した検索条件の設定を行っていますが、この場合、比較演算子をダブルクォーテーションで囲った上で、「&」記号で参照先の G4 と接続して検索条件を構成します。

=COUNTIF(D2:D12,">"&G4)

図 6. 図 3 に入力されている集計関数の式

	F	G	H
1	COUNTIF		
2	旅費		=COUNTIF(C2:C12,"旅費")
3	金額 > 3万		=COUNTIF(D2:D12,">30000")
4	金額 >	30000	=COUNTIF(D2:D12,">"&G4)
5			
6	SUMIF		
7	旅費		=SUMIF(C2:C12,F7,D2:D12)
8			
9		COUNTIFS	SUMIFS
10	旅費	件数	金額
11	営業部	=COUNTIFS(\$C\$2:\$C\$12,\$F\$10,\$B\$2:\$B\$12,F11)	=SUMIFS(\$D\$2:\$D\$12,\$C\$2:\$C\$12,\$F\$10,\$B\$2:\$B\$12,F11)
12	総務部	=COUNTIFS(\$C\$2:\$C\$12,\$F\$10,\$B\$2:\$B\$12,F12)	=SUMIFS(\$D\$2:\$D\$12,\$C\$2:\$C\$12,\$F\$10,\$B\$2:\$B\$12,F12)

出所：AFP(2024) Exhibit15-4 を参考に筆者作成

セル H7 では、C2:C12 の使途の中で F7 に入力された旅費に該当する行を抽出し、その行の D2:D12 に含まれる値を合計しています。

=SUMIF(C2:C12,F7,D2:D12)

セル G11 では、以下の式により、使途が旅費、部署が営業部のデータの件数を返しています。

=COUNTIFS(\$C\$2:\$C\$12,\$F\$10,\$B\$2:\$B\$12,F11)

検索条件範囲 1 と検索条件 1 を絶対参照にすることで、使途を旅費に固定し、検索条件範囲 2 を絶対参照、検索条件 2 は相対参照とすることで、G12 へオートフィルした際には部署が総務部に切り替わり、G11 で営業部の旅費、G12 で総務部の旅費の件数を返しています。

セル H11 では、以下の式により、営業部の旅費の合計金額を算出しています。

=SUMIFS(\$D\$2:\$D\$12,\$C\$2:\$C\$12,\$F\$10,\$B\$2:\$B\$12,F11)

ここでも、合計範囲と検索条件範囲 1、検索条件 1、検索条件範囲 2 を絶対参照として、検索条件 2 のみ相対参照とすることで、H12 へのオートフィルで総務部の合計金額を返すことが可能になっています。

2. 配列数式

AFP(2024)で示されているスプレッドシートの活用法の特徴のひとつが、配列数式が積極的に活用されていることです。

Microsoft は、「配列は、値の行または列、または値の行と列の組み合わせと考えることができます。配列数式は、複数の結果または 1 つの結果を返すことができます。」として、「配列内の 1 つ以上の項目に対して複数の計算を実行できる数式です。」と説明しています²。

配列には、単一セルの配列数式 (Single cell arrays)、複数セルの配列数式 (Multicell arrays)、複数の行・列に値を入力したり数式の定数として使用できる配列定数 (Array constants) があります。

配列の使用例と操作方法を、AFP(2024, Topic 1 15-13)は以下の様に説明しています。

例えば、100 行あって、各行で異なる 2 つの列の積を求める場合、単純に=\$A1*\$B1 といった具合に行ごとにすればよいかもしれませんが、結果が表示される列の範囲を選択して、範囲を含む数式を追加することで、配列数式を作成することもできます。例えば、範囲 C1:C100 を選択して、数式バーに=A1:A100*B1:B100 と入力し、Ctrl+Shift+Enter を押すと、配列数式 (CSE 数式³とも呼ばれる) が作成されます。配列は自動的に追加される {} 括弧で示されます。Ctrl+Shift+Enter は必須です。括弧を手動で追加しても配列にはなりません。

なお、「Excel 365 の動的配列数式機能は、隣接するセルに自動的にスピルし、レガシーな CSE の組み合わせを不要にします。」 (AFP, 2024, Topic 1 15-13)とされているとおり、現在提供されている Microsoft 365 の Excel では、結果を出力する範囲の左上のセルに数式を入力して Enter キーを押すことで、数式を動的配列数式として入力することができるようになりました。

² Microsoft サポート「配列数式のガイドラインと例」. <https://support.microsoft.com/ja-jp/office/%E9%85%8D%E5%88%97%E6%95%B0%E5%BC%8F%E3%81%AE%E3%82%AC%E3%82%A4%E3%83%89%E3%83%A9%E3%82%A4%E3%83%B3%E3%81%A8%E4%BE%8B-7d94a64e-3ff3-4686-9372-ecfd5caa57c7>

³ 筆者注：CSE は「Ctrl Shift Enter」を意味する。

(1) 配列数式の利点

AFP(2024, Topic 1 15-13)は、配列を一連の数式の代わりに使用する目的として、以下の 4 点を挙げています。

- ✓ すべての計算が同一であることを確実にするため
- ✓ 複雑な数式に誤りがある可能性を減らすため
- ✓ 範囲内の 1 つ以上の数式が上書きされたり、改ざんされたりするのを防ぐため
- ✓ 他の計算などに影響を及ぼす可能性がある場合に、配列範囲の行や列の追加や削除を防ぐため

(2) 配列数式の使用例

図 7. 配列数式の例

	A	B	C	D	E
1	フレーバー	ユニット売上	価格/ユニット	計	
2	コーヒー	98	¥650	¥63,700	=B2:B4*C2:C4
3	アップル	120	¥520	¥62,400	
4	ミルク	64	¥390	¥24,960	
5	売上			¥151,060	=SUM(B2:B4*C2:C4)
6				¥151,060	=SUMPRODUCT(B2:B4,C2:C4)

出所：AFP(2024) Exhibit15-7, Exhibit15-8 を参考に筆者作成

図 7 は、配列関数の使用例です。フレーバーごとの売上合計を算出する際、一般的には先頭行 (D2) にユニット売上×価格/ユニット (=B2*C2) と入力して D4 までオートフィルして埋めますが、セル D2 に=B2:B4*C2:C4 と入力して Enter を押すことで動的複数セル配列により D2:D4 で B 列×C 列の計算を行うことができます。

セル D5 では単一セルの配列として=SUM(B2:B4*C2:C4)とすることで、=SUM(B2*C2,B3*C3,B4*C4)や、配列間の対応する要素の積を返す=SUMPRODUCT(B2:B4,C2:C4)と同じ計算結果を返すことができます。

(3) SUM と HLOOKUP を配列として使用する

AFP(2024, Topic 1 15-15)では、配列の活用法として、SUM 関数と検索関数の HLOOKUP 関数を配列として使用方法が例示されています。HLOOKUP 関数は、

=HLOOKUP(検索値, 範囲, 行番号, 検索方法)

で構成され、範囲の 1 行目と検索値を指定された検索方法で照合して対応する列を決定した後、行番号が指定する行の値を返す関数です。

配列を使用しない場合、=SUM(HLOOKUP(B9,B2:H6,2,0), HLOOKUP(B9,B2:H6,3,0), HLOOKUP(B9,B2:H6,4,0), HLOOKUP(B9,B2:H6, 5,0))のように、検索値に対応する列（曜日）の各行（店舗）の数値を4つのHLOOKUP関数で抽出して合計する必要があります。

SUM関数にネストしたHLOOKUP関数の行番号の引数に配列を使用し、{2,3,4,5}と指定することにより、検索値に対応する曜日の4店舗の数値を合計することが可能になります。

図 8. 配列としてのSUM関数とHLOOKUP関数の使用例

	A	B	C	D	E	F	G	H
1								
2	支店名	月	火	水	木	金	土	日
3	東京	788	618	814	740	391	521	904
4	御茶ノ水	831	610	433	365	745	369	532
5	四ツ谷	575	545	652	452	450	781	958
6	新宿	931	655	929	611	620	484	800
7		3,125	2,428	2,828	2,168	2,206	2,155	3,194
8								
9	曜日	水						
10	合計	2,828	=SUM(HLOOKUP(B9,B2:H6,{2,3,4,5},0))					

出所：AFP(2024) Exhibit15-9 を参考に筆者作成

(4) TRANSPOSE 関数

配列の便利な活用法として、引数に指定した範囲の行列を入れ替える TRANSPOSE 関数が挙げられます AFP(2024, Topic 1 15-16 to 15-17)。

Excel では、表を選択してコピーし、行列を入れ替えて貼り付けることができますが、この方法ではコピー元の表とはリンクされない、ハードコードされた値になってしまいます。

TRANSPOSE 関数を用いることで、元の値を参照した状態で行と列を入れ替えることができるため、値を修正する必要がある場合も、参照元のセルのみを修正すれば行列を入れ替えた先にも変更が反映され、効率的に作業ができ、エラーを回避することにも繋がります。

図 9 では、セル A12 に=TRANSPOSE(A2:H6)と入力して Enter を押すことで、動的配列により A12:E19 に、A2:H6 の行と列を入れ替えた値が返されます⁴。

⁴ 従来は、貼り付け先にもコピー元の行数・列数を入れ替えた行数・列数を指定した上で TRANSPOSE 関数を入力し、Ctrl+Shift+Enter で確定する必要がありましたが、現在は貼り付け先の左上のセルのみを選択して TRANSPOSE 関数を入力し、Enter で動的配列を入力できるため、利便性が大幅に向上しました。

図 9. TRANSPOSE 関数の使用例

A12		=TRANSPOSE(A2:H6)						
	A	B	C	D	E	F	G	H
1								
2	支店名	月	火	水	木	金	土	日
3	東京	788	618	814	740	391	521	904
4	御茶ノ水	831	610	433	365	745	369	532
5	四ツ谷	575	545	652	452	450	781	958
6	新宿	931	655	929	611	620	484	800
7		3,125	2,428	2,828	2,168	2,206	2,155	3,194
8								
9	曜日	水						
10	合計	2,828						
11								
12	支店名	東京	御茶ノ水	四ツ谷	新宿			
13	月	788	831	575	931			
14	火	618	610	545	655			
15	水	814	433	652	929			
16	木	740	365	452	611			
17	金	391	745	450	620			
18	土	521	369	781	484			
19	日	904	532	958	800			

出所：AFP(2024) Exhibit15-9 を参考に筆者作成

3. 検索関数

AFP(2024)は、Excel の検索関数は財務分析に必須のツールであるとして、これらの関数を活用することにより、FP&A プロフェッショナルは、財務データを効率的に取得し、操作することができるとしています (Topic 4 15-32)。

AFP(2024, Topic 4 15-32)は、論理関数の IF を使用することである種の検索を行うことは可能であるが、複数の IF をネストする必要があるような複雑なシナリオを扱う場合、検索関数を用いることでより効率的に作業ができるとしています。

(1) VLOOKUP 関数

Excel の関数の中で SUM 関数に次いで使用頻度が高いとされているのが VLOOKUP 関数です。

VLOOKUP 関数は、

=VLOOKUP(検索値, 範囲, 列番号, 検索方法)

で構成され、検索値を範囲の一行目と照合して垂直方向 (Vertical) に検索 (Lookup) し、行を確定した上で、列番号が指定した列の値を返します。検索方法に TRUE もしくは 1 を指定した場合、近似

検索（厳密には、範囲の 1 列目の値以上が、未満かによって行を確定）が行われ、FALSE もしくは 0 が指定された場合には完全一致検索が行われます。VLOOKUP では標準で近似検索が指定されるため、TRUE もしくは 1 は省略することができます。

HLOOKUP 関数は、垂直方向（Vertical）に検索する VLOOKUP 関数に対して、検索する方向が水平方向（Horizontal）になる以外は VLOOKUP と同じ働きをします。

図 10. 検索関数の活用例

	A	B	C	D
1	売上総額の範囲			
2	下限	上限	割引率	
3	¥0	¥100,000	0%	
4	¥100,001	¥250,000	2%	
5	¥250,001	¥500,000	3%	
6	¥500,001	¥750,000	4%	
7	¥750,001	¥1,000,000	5%	+
8	¥1,000,001		6%	
9				
10				
11	売上総額の入力	¥451,000		
12	割引率 (VLOOKUP)	3%	=VLOOKUP(B11,A3:C8,3)	
13	割引率 (XLOOKUP)	3%	=XLOOKUP(B11,A3:A8,C3:C8,,-1)	
14				
15	割引後売上高	¥437,470	=B11*(1-B12)	

出所：AFP(2024) Exhibit15-16 を参考に筆者作成

図 10 は、売上総額によって割引率が変動することを想定したシートです。近似検索を行う場合、VLOOKUP 関数で検索値と照らし合わせる範囲の 1 列目は昇順に配置され、該当する範囲の最小値が入力されている必要があります。

図 10 のセル B12 では、VLOOKUP 関数で B11 に入力した売上総額を A3:C8 の 1 列目すなわち A3:A8 と照合して検索を行い、近似検索で該当した行の 3 列目すなわち C 列の割引率を返しています。

(2) XLOOKUP 関数

VLOOKUP 関数、HLOOKUP 関数の後継として登場したのが XLOOKUP 関数です。

XLOOKUP 関数は以下の様に構成され、見つからない場合、一致モード、検索モードは省略可能です。

=XLOOKUP(検索値, 検索範囲, 戻り配列, 見つからない場合, 一致モード, 検索モード)

VLOOKUP 関数では、検索を行う列と結果として返す列をまとめて「範囲」として指定し、結果として返す列を列番号で指定するのに対し、XLOOKUP 関数では、検索を行う列または行（検索範囲）と結果として返す列または行（戻り範囲）を独立して指定できるようになりました。また、4つ目の引数「見つからない場合」には、検索結果が見つからない場合にエラーの代わりに返す要素を指定することができます。5つ目の引数、「一致モード」は、VLOOKUP 関数の検索方法に相当し、一致の種類を指定します。

表 1. XLOOKUP 関数の一致モード

指定の方法	一致の種類
0	完全一致。見つからない場合は、#N/A が返されます。※既定の設定
-1	完全一致。見つからない場合は、次の小さなアイテムが返されます。
1	完全一致。見つからない場合は、次の大きなアイテムが返されます。
2	＊、?、および ~ が特別な意味を持つワイルドカードの一致。

出所：Microsoft サポート「XLOOKUP 関数」⁵より

6つ目の引数、「検索モード」は、先頭、末尾のどちらから検索するか、あるいはバイナリ検索を行うかを指定します。

表 2. XLOOKUP 関数の検索モード

指定の方法	検索モード
1	先頭の項目から検索を実行します。これが既定の設定です。
-1	末尾の項目から逆方向に検索を実行します。
2	昇順で並べ替えられた検索範囲を使用してバイナリ検索を実行します。並べ替えられていない場合、無効な結果が返されます。
-2	降順で並べ替えられた検索範囲を使用してバイナリ検索を実行します。並べ替えられていない場合、無効な結果が返されます。

出所：Microsoft サポート「XLOOKUP 関数」⁶より

図 10 のセル B13 では、XLOOKUP 関数により割引率を算出しています。検索値に B11、検索範囲に A3:A8 が指定され、戻り範囲には C3:C8 が指定されています。「見つからない場合」は省略され、一致モードには-1 が指定されることで、¥250,001 以上¥500,001 未満に該当し、戻り範囲から 3%が返されています。

⁵ Microsoft サポート「XLOOKUP 関数」. <https://support.microsoft.com/ja-jp/office/xlookup-%E9%96%A2%E6%95%B0-b7fd680e-6d10-43e6-84f9-88eae8bf5929>

⁶ 同上

図 1 1. SUM 関数と XLOOKUP 関数の活用例

	A	B	C	D	E	F	G	H	I	J	K
1	日付	借方	貸方								
2	4月15日	6,135	6,587		開始日	4月16日					
3	4月16日	7,038	7,490		終了日	4月20日					
4	4月17日	7,942	8,394		借方計	44,226	=SUM(XLOOKUP(F2,A2:A11,B2:B11):XLOOKUP(F3,A2:A11,B2:B11))				
5	4月18日	8,845	9,297		貸方計	46,486	=SUM(XLOOKUP(F2,A2:A11,C2:C11):XLOOKUP(F3,A2:A11,C2:C11))				
6	4月19日	9,749	10,201								
7	4月20日	10,652	11,104		借方計	44,226	=SUMIFS(B2:B11,A2:A11,">="&F2,A2:A11,"<="&F3)				
8	4月21日	11,556	12,008		貸方計	46,486	=SUMIFS(C2:C11,A2:A11,">="&F2,A2:A11,"<="&F3)				
9	4月22日	12,459	12,911								
10	4月23日	13,363	13,815								
11	4月24日	14,266	14,718								

出所：AFP(2024) Exhibit15-17 を参考に筆者作成

図 1 1 は借方・貸方それぞれで XLOOKUP 関数によって開始日と終了日の値を抽出し、SUM 関数にネストすることで、SUMIF 関数と同等の機能を実現している例です。

(3) OFFSET 関数

OFFSET 関数は、セルまたはセル範囲から指定された行数と列数だけシフトした位置にあるセル範囲の参照を返します。返されるセル参照は、単一のセル、セル範囲のいずれかで、返されるセル参照の行数と列数を指定することもできます。⁷

OFFSET 関数は、以下の様に構成されます。

=OFFSET(基準, 行数, 列数, [高さ], [幅])

OFFSET 関数は、式が入力されているセルから行数と列数の数だけ移動します。値が正の場合は右や下に、値が負の場合は左や上に移動します。

セル B7 には

=OFFSET(C7,0,B4,5)

が入力されています。セル C7 を基準に、行はそのまま、セル B4 に入力した値の数だけ列を移動し、5 行分の高さを取得することで、選択したシナリオの初期費用から耐用年数（年）までの値を取得します。B4 に 0 を入力すると基本シナリオが、1 を入力するとベストのシナリオが、2 を入力するとワーストのシナリオが返されます。

⁷ Microsoft サポート「OFFSET 関数」. <https://support.microsoft.com/ja-jp/office/offset-%E9%96%A2%E6%95%B0-c8de19ae-dd79-4b9b-a14e-b4d906d11b66>

図 1 2. OFFSET 関数の活用例

	A	B	C	D	E
1	設備投資に関連するCFの簡易評価				
2					
3	インプット				
4	データセット	1			
5			0	1	2
6	シナリオ		基本	ベスト	ワースト
7	初期費用	1,800,000	2,000,000	1,800,000	2,800,000
8	残存価値	200,000	80,000	200,000	0
9	税引後年間営業CF	1,000,000	600,000	1,000,000	400,000
10	割引率	8%	10%	8%	12%
11	耐用年数 (年)	8	8	8	8
12					
13		セルB7の式: =OFFSET(C7,0,B4,5)			

出所：AFP(2024) Exhibit15-18 を参考に筆者作成

(4) MATCH 関数

MATCH 関数は、検査範囲の中での検査値の相対的な位置を返す関数です。

MATCH 関数は以下の様に構成されます。

=MATCH(検査値, 検査範囲, [照合の型])

MATCH 関数の検査範囲には単一の行もしくは列を指定する必要があり、照合の型には 0, 1, もしくは-1 を指定します。

表 3. MATCH 関数の照合の型

照合の型	動作
1 または省略	MATCH 関数は、検査値以下の最大の値を検索します。検査範囲の引数の値は、昇順の並べ替えでは、1 ~ 9、A ~ Z、あ~ん、FALSE ~ TRUE の順に配置されます。
0	MATCH 関数は、検査値と等しい最初の値を検索します。検査範囲の引数の値は、任意の順序で指定できます。
-1	MATCH 関数は、検査値以上の最小値を検索します。検査範囲の引数の値は、TRUE、FALSE、Z-A、..などの降順で配置する必要があります。2、1、0、-1、-2、..など。

出所：Microsoft サポート「MATCH 関数」⁸より

⁸ Microsoft サポート「MATCH 関数」<https://support.microsoft.com/ja-jp/office/match-%E9%96%A2%E6%95%B0-e8dff45-c762-47d6-bf89-533f4a37673a>

図 1 3. OFFSET 関数と MATCH 関数の活用例

	A	B	C	D	E	F
1	設備投資に関するCFの簡易評価					
2						
3	インプット					
4	シナリオ	ワースト				
5	データセット	3	=MATCH(B6,C10:E10,0)			
6						
7			1	2	3	
8			基本	ベスト	ワースト	
9	初期費用	2,800,000	2,000,000	1,800,000	2,800,000	
10	残存価値	0	80,000	200,000	0	
11	税引後年間営業CF	400,000	600,000	1,000,000	400,000	
12	割引率	12%	10%	8%	12%	
13	耐用年数 (年)	8	8	8	8	
14						
15	セルB9の式：		=OFFSET(C9,0,B5-1,5)			

出所：AFP(2024) Exhibit15-19 を参考に筆者作成

図 1 3 は、OFFSET 関数と MATCH 関数を組み合わせてシナリオを選択する方法を示しています。セル B4 には、基本、ベスト、ワーストの 3 つのシナリオのいずれかを入力します。セル B5 には MATCH 関数が以下の様に入力されています。

=MATCH(B4,C8:E8,0)


この MATCH 関数により、セル B5 には B4 で選択したシナリオが B8:E8 で左から何列目に一致するかが返されます。セル B5 の MATCH 関数の結果を受け、セル B9 の

=OFFSET(C9,0,B5-1,5)

の OFFSET 関数により、B9:B13 に、選択したシナリオに対応する初期費用から耐用年数 (年) の値が返されます。シナリオ 1 が基本のシナリオということになりますので、セル B5 に 1 が返ってきたときにセル B9 の OFFSET 関数の列数が 0 になるよう、列数の引数は B5-1 としています。

図 1 3 では、セル B4 にシナリオ名を入力しますが、これを正確に入力しないとエラーになり、一連の関数は正しく機能しません。この問題の解決策として、データの入力規則を活用する方法があります (AFP, 2024, Topic 4 15-38)。セル B4 をアクティブにしてデータの入力規則を開き、入力値の種類にリストを選択します。元の値として、基本、ワースト、ベストのシナリオ名が入力されている=C8:E8 を指定します。元の値には絶対参照形式 (=\$C\$8:\$E\$8) で格納され、基本、ワースト、ベストをドロップダウンリストから選択できるようになります。

図 14. データの入力規則の設定



出所：AFP(2024) Exhibit15-20 を参考に筆者作成

OFFSET 関数も MATCH 関数も、基本的な機能は単純であり明解ですが、個々の関数の引数の組み合わせと機能を理解した上で、適切に組み合わせることにより、極めて高い利便性を提供します。

なお、MATCH 関数の後継として登場した XMATCH 関数は、

=XMATCH(検査値, 検査範囲, 一致モード, 検索モード)

で構成され、一致モード、検索モードは省略可能です。一致モードにはワイルドカード検索の「2」が追加されています。

Microsoft 365 では、XLOOKUP 関数や XMATCH 関数のように、500 近く存在する関数の中でも最も主要な関数の後継となる関数が次々と登場しています。5 年ぶりに更新された AFP(2024)では、これらの新しい関数や機能の活用にも極めて意欲的に取り組まれています。このことは、スプレッドシートの代表的アプリケーションとしての Excel のアップデートが、今後も FP&A プロフェッショナルの業務を拡張していく可能性を示唆していると言えるのではないのでしょうか。

IV. おわりに

本稿では、FP&A プロフェッショナルに求められるスプレッドシートの基本操作と主要な関数について、具体的な事例を交えて解説しました。会計や財務分析の分野において、スプレッドシートは依然として重要なツールであり、その活用スキルは意思決定の精度を高める上で欠かせないものであるといえます。

企業の経理部門や財務分析を担う部門では、データの整理、分析、共有が日常的に求められますが、スプレッドシートはそのすべてのプロセスを支える基盤として機能します。本文で取り上げた数式や関

数、配列数式の活用法は、業務の効率化や正確性の向上に直結します。たとえば、配列数式による一括処理や検索関数を用いたデータ抽出の効率化は、膨大なデータを取り扱う場面で特に威力を発揮します。さらに、最新の Excel 機能である動的配列や XLOOKUP 関数の活用により、従来の操作に比べて柔軟性や効率性が大幅に向上しています。

また、スプレッドシートのスキルを習得することは、BI ツールや ERP システムの活用にも役立ちます。これらの高度なツールを用いる場合でも、最終的にはスプレッドシートで加工・集約し、他者と共有することが多いためです。特に、社内外の関係者に情報をわかりやすく提示し、迅速な意思決定を支援するためには、スプレッドシートでの整理と可視化が重要となります。

会計や経理の専門職としてキャリアを築く皆様にとって、スプレッドシートスキルの習得は単なる技術の取得にとどまりません。それは、分析力や説得力を備えたプロフェッショナルとしての成長を支える基盤となります。本稿が、皆様の日々の業務やスキルアップの参考となり、今後のご活躍に役立つことを心より願っています。

筆者自身もまた、会計実務におけるスプレッドシートの可能性をさらに探究し続ける所存です。新たな知見を得て皆様と共有できる日を楽しみにしながら、引き続き研究を深めてまいります。スプレッドシートという普遍的なツールが、会計実務をさらに発展させる原動力となるよう、共に学びを進めていきましょう。

(参考文献)

Association for Financial Professionals, 2024, FPAC Exam Prep Platform: Chapter 15 Spreadsheet Functions.

Microsoft サポート「MATCH 関数」<https://support.microsoft.com/ja-jp/office/match-%E9%96%A2%E6%95%B0-e8dff45-c762-47d6-bf89-533f4a37673a>

Microsoft サポート「OFFSET 関数」. <https://support.microsoft.com/ja-jp/office/offset-%E9%96%A2%E6%95%B0-c8de19ae-dd79-4b9b-a14e-b4d906d11b66>

Microsoft サポート「XLOOKUP 関数」. <https://support.microsoft.com/ja-jp/office/xlookup-%E9%96%A2%E6%95%B0-b7fd680e-6d10-43e6-84f9-88eae8bf5929>

Microsoft サポート「配列数式のガイドラインと例」. <https://support.microsoft.com/ja-jp/office/%E9%85%8D%E5%88%97%E6%95%B0%E5%BC%8F%E3%81%AE%E3%82%AC%E3%82%A4%E3%83%89%E3%83%A9%E3%82%A4%E3%83%B3%E3%81%A8%E4%BE%8B-7d94a64e-3ff3-4686-9372-ecfd5caa57c7>